


# Training Script Language PRACTICE

---

[TRACE32 Online Help](#)

[TRACE32 Directory](#)

[TRACE32 Index](#)

|   |   |
|---|---|
| <b>TRACE32 Training</b> .....                         |  |
| <b>Training Script Language PRACTICE</b> .....        | <b>1</b>  |
| <b>History</b> .....                                  | <b>3</b>  |
| <b>E-Learning</b> .....                               | <b>3</b>  |
| <b>Ready-to-Run Scripts</b> .....                     | <b>3</b>  |
| <b>Introduction to Script Language PRACTICE</b> ..... | <b>4</b>  |
| Area of Use   | 4   |
| Run a Script  | 5   |
| <b>Create a PRACTICE Script</b> .....                 | <b>6</b>  |
| Convert TRACE32 Settings to a Script                  | 6   |
| Command LOG   | 11  |
| Command History                                       | 12  |
| Script Editor PEDIT                                   | 13  |
| Syntax Highlighting                                   | 15  |
| <b>Debugging of PRACTICE Script</b> .....             | <b>17</b>   |
| Debug Environment                                     | 18  |
| Display the PRACTICE Stack                            | 21  |
| <b>First Examples</b> .....                           | <b>22</b>   |
| Run Through Program and Generate a Test Report        | 22  |
| Check Contents of Addresses                           | 28  |
| Check Contents of Address Range                       | 29  |
| Check the Contents of Variables                       | 32  |
| Record Formatted Variables                            | 33  |
| Record Variable as CSV                                | 34  |
| Test Functions  | 36  |
| Test Function with Parameter File                     | 37  |
| <b>Structure of PRACTICE Programs</b> .....           | <b>39</b>   |
| Program Elements                                      | 39  |
| Commands  | 39  |
| Functions   | 40  |
| Comments  | 42  |
| Labels  | 42  |
| Program Flow Control                                  | 44  |

|                               |           |
|-------------------------------|-----------|
| Conditional Program Execution | 49        |
| PRACTICE Macros               | 51        |
| Create a Macro                | 51        |
| Assign Content to a Macro     | 53        |
| Macro Handling                | 55        |
| Macros as Strings             | 57        |
| Macros as Numbers             | 59        |
| Note for Testing              | 62        |
| More Complex Data Structures  | 63        |
| I/O Commands                  | 64        |
| File Commands                 | 70        |
| Event Control via PRACTICE    | 72        |
| <b>Index (local)</b> .....    | <b>73</b> |

## History

---

The most current version of this training manual can be downloaded from:

<http://www.lauterbach.com/training.html>

31-Jul-15      Partly revised.

## E-Learning

---

Videos about the script language PRACTICE can be found here:

[http://www.lauterbach.com/tut\\_practice.html](http://www.lauterbach.com/tut_practice.html)

## Ready-to-Run Scripts

---

Ready-to-run PRACTICE scripts provided by the Lauterbach experts are published and updated daily here:

<http://www.lauterbach.com/scripts.html>

## Area of Use


---

The main tasks of PRACTICE scripts are:

- to provide the proper start-up sequence for the development tool
- to automate FLASH programming
- to customize the user interface
- to store and reactivate specific TRACE32 settings
- to run automatic tests

The standard extension for PRACTICE scripts is **.cmm**.

# Run a Script



```
CD.DO *                                // "*" opens a file browser for script
                                        // selection

                                        // TRACE32 first changes to the directory
                                        // where the selected script is located and
                                        // then starts the script
```

|                            |  |
|----------------------------|--|
| <b>ChDir.DO</b> <filename> | Change to the directory where the script <filename> is located and start the script. |
| <b>DO</b> <filename>       | Start script <filename>.   |
| <b>PATH</b> [+<path_name>  | Define search paths for PRACTICE scripts.  |

```
DO memtest

ChDir.DO c:/t32/demo/powerpc/hardware

PATH c:/t32/tests
```


## Convert TRACE32 Settings to a Script

The commands **STOre** and **ClipSTOre** generate scripts that allow to reactivate the specified TRACE32 *<setting>* at any time.

*<setting>* is in most cases the set-up of a command group.

|                        |  |
|------------------------|--|
| <i>&lt;setting&gt;</i> |  |
| <b>SYStem</b>          | Setting for command group SYStem.          |
| <b>Break</b>           | Setting for command group Break.           |
| <b>Win</b>             | TRACE32 window configuration.              |
| <b>NoDate</b>          | Do not include time stamp into the script. |

|   |  |
|---|--|
| <b>ClipSTOre</b> { <i>&lt;setting&gt;</i> }                     | TRACE32 creates a script that allows to reactivate the selected settings. The script is stored to the clipboard. |
| <b>STOre</b> <i>&lt;filename&gt;</i> { <i>&lt;setting&gt;</i> } | TRACE32 creates the script <i>&lt;filename&gt;</i> to reactivate the selected settings.                          |




ClipSTore SYStem NoDate

```

B::
SYSTEM.RESET
SYSTEM.CPU SPC56EC74
SYSTEM.CONFIG CORENUMBER 2.
SYSTEM.CONFIG CORE 1. 1.
CORE.ASSIGN 1.
SYSTEM.MEMACCESS NEXUS
SYSTEM.CPUACCESS DENIED
SYSTEM.OPTION IMASKASM OFF
SYSTEM.OPTION IMASKHLL OFF
SYSTEM.BDMCLOCK 4000000.
SYSTEM.CONFIG TRISTATE OFF
SYSTEM.CONFIG SLAVE OFF
SYSTEM.CONFIG TAPSTATE 7.
SYSTEM.CONFIG TCKLEVEL 0.
SYSTEM.CONFIG.DEBUGPORT Analyzer0
SYSTEM.CONFIG CJTAGFLAGS 0x3
SYSTEM.MODE UP

ENDDO
    
```




ClipStore Break NoDate

```
B::  
  
BREAK.RESET  
B.S func24 /P  
B.S main\26 /P  
B.S sieve /P /O /COUNT 1000.  
V.B.S vfloat; /W  
  
ENDDO
```

The breakpoints are saved at a symbolic level by default.





```
STOre window_configuration.cmm Win NoDate           // TRACE32 creates script
                                                       // window_configuration
                                                       // to reactivate the
                                                       // current window
                                                       // configuration
```

B::

TOOLBAR ON

STATUSBAR ON

FramePOS 15.625 8.9286 193. 47.

WinPAGE.RESet

WinPAGE.Create P000

WinCLEAR

WinPOS 0.0 22.214 80. 5. 0. 0. W002

Var.View %SpotLight.on %E flags %Open vtripplearray

WinPOS 0.0 31.429 80. 8. 5. 0. W003

Frame /Locals /Caller

WinPOS 0.0 0.0 80. 16. 13. 1. W000

WinTABS 10. 10. 25. 62.

List.auto

WinPOS 84.25 0.0 77. 20. 0. 0. W004

Register.view

WinPOS 83.875 24.071 105. 6. 0. 0. W001

PER , "FlexCAN"

WinPAGE.select P000

ENDDO

# Command LOG

The LOG command allows to record most of the activities in the TRACE32 PowerView GUI.

Commands to control the command LOG:

|                        |  |
|------------------------|--|
| <b>LOG.OPEN</b> <file> | Create and open a file for the command LOG. The default extension for LOG files is (.log). |
| <b>LOG.CLOSE</b>       | Close the command LOG file.  |
| <b>LOG.OFF</b>         | Switch off command LOG temporarily.  |
| <b>LOG.ON</b>          | Switch on command LOG.   |
| <b>LOG.type</b>        | Display command LOG while recording.   |

|                 |   |
|-----------------|---|
| LOG.OPEN my_log | Creates and opens the .log file             |
| TYPE my_log.log | Displays .log file contents while recording |
| ...             | Recording                                   |
| LOG.CLOSE       | Closes .log file                            |

Contents of a command-LOG

```
B::B::List
B::Go func24
// B::LOG.ON
B::B::PER , "Analog to Digital Converter"
B::B::PER.Set.simple ANC:0xFFE00000 %L (d.l(ANC:0xFFE00000)&~0x40000000)|0x40000000
```

# Command History



The command history records only commands entered into the command line.  
The default extension for HISTory-files is (**.log**)

## HISTory.type

Display the command history

```
B::HISTory
B::LOG.OPEN my_log
B::Go func24
B::log.off
B::LOG.ON
B::LOG.CLOSE
B::log.type
B::TYPE my_log.log
B::history.save my_history
B::dir
B::HISTory
```

## HISTory.SAVE [<filename>]

Save the command history

## HISTory.SIZE [<size>]

Define the size of the command history

If the file **T32.cmm** contains the instruction:


```
AutoSTOre , HISTory
```

then the command history is automatically saved in the TMP directory at the exit of TRACE32 and recalled when TRACE32 is started.

## AutoSTOre <filename> {<setting>}

Store defined settings automatically at the exit of TRACE32 and reactivate them at the start of TRACE32

# Script Editor PEDIT



**CD.PEDIT \***

```
// "*" opens a file browser for script  
// selection  
  
// TRACE32 first changes to the directory  
// where the selected script is located and  
// then opens the script in a PEDIT window
```

**ChDir.PEDIT** <filename>

Change to the directory where the script <filename> is located and open script in script editor PEDIT.

**PEDIT** <filename>

Open script <filename> in script editor PEDIT.

Save PRACTICE script

Save PRACTICE script with a new name


Close and save PRACTICE script

Close and do not save PRACTICE script

Save and then start PRACTICE script

Start PRACTICE script

Debug PRACTICE script



# Syntax Highlighting

The script editor PEDIT, unfortunately does not support syntax highlighting for PRACTICE scripts.

If you want to have syntax highlighting for PRACTICE scripts, you have to use the following workaround.

1. Redirect the call of the TRACE32 editor EDIT to an external editor by using the command TRACE32 command **SETUP.EDITTEXT**.
2. Install syntax highlighting files provided by Lauterbach for the external editor.

**EDIT** *<filename>*

Open file with standard TRACE32 editor.

**SETUP.EDITTEXT ON** *<command>*

Advise TRACE32 to use the specified external editor if the EDIT command is used.

*<command>* contains the command that TRACE32 sends to your host OS to start the external editor. In this string the following replacements will be made:


- \* will be replaced by the actual file name.
- # will be replaced by the actual line number.

Lauterbach provides syntax highlighting files for some common text editors. Please refer to

`~~\demo\practice\syntaxhighlighting` for details. `~~` stands for the

`<TRACE32_installation_directory>`, which is `c:/T32` by default.

```
// Advise TRACE32 to use TextPad when the EDIT command is used
SETUP.EDITTEXT ON "C:\Program Files (x86)\TextPad 5\TextPad.exe "*" (#)""
```






TRACE32 supports the debugging of PRACTICE programs.


A short video that provide an introduction into PRACTICE debugging is available on:

[http://www.lauterbach.com/tut\\_practice.html](http://www.lauterbach.com/tut_practice.html)

Use **Debug** to start the debugger for the PRACTICE script



```
Go main
WAIT !STATE.RUN() 2.s
IF STATE.RUN()
(
  Break.direct
)
IF Register(PC)==ADDRESS.OFFSET(main)
(
  APPEND test_protocol.txt FORMAT.STRING("System booted successfully",35,' ') FORMAT.STRING(STATE.TIME(),12,' ')
)
ELSE
(
  APPEND test_protocol.txt FORMAT.STRING("Booting failed",35,' ') FORMAT.STRING(STATE.TIME(),12,' ')
)
```



```
1 DO ~\demo\powerpc\hardware\spc56xx\spc564bc\target_setup.cmm
3 Go main
5 WAIT !STATE.RUN() 2.s
7 IF STATE.RUN()
8 (
9   Break.direct
10 )
12 IF Register(PC)==ADDRESS.OFFSET(main)
13 (
14   APPEND test_protocol.txt FORMAT.STRING("System booted successfully",35,' ') FORMAT.STRING(STATE.TIME(),12,' ')
15 )
16 ELSE
17 (
18   APPEND test_protocol.txt FORMAT.STRING("Booting failed",35,' ') FORMAT.STRING(STATE.TIME(),12,' ')
19 )
20 ENDDO
```

## WinResist.WinExt.ChDir.PLIST

**WinResist:** PEDIT window is not deleted by command **WinCLEAR**.

**WinExit:** Detach PEDIT window from the TRACE32 main window - even if TRACE32 is operating in MDI window mode.


**PSTEP** <filename>

Start script in PRACTICE debugger.


**ChDir.PSTEP** <filename>

TRACE32 first changes to the directory where the script is located and then starts the script in the PRACTICE debugger.

**WinResist.WinExt.ChDir.PSTEP** <filename>



| <i>Local buttons in PLIST/PSTEP window</i> |  |
|--|--|
| <b>Step</b>                                | Single step PRACTICE script (command <b>PSTEP</b> ).   |
| <b>Over</b>                                | Run called PRACTICE script or PRACTICE subroutine as a whole (command <b>PSTEPOVER</b> ).  |
| <b>Up</b>                                  | End current PRACTICE script or subroutine and return to the caller (command <b>PSTEPDOWN</b> ).  |
| <b>Continue</b>                            | Continue the execution of PRACTICE script (command <b>CONTINUE</b> ).  |
| <b>Stop</b>                                | Stop the execution of the PRACTICE script (command <b>STOP</b> ).  |
| <b>Enddo</b>                               | End the current PRACTICE script. Execution is continued in the calling PRACTICE script. If no calling script exists, the PRACTICE script execution is ended (command <b>ENDDO</b> ). |
| <b>List</b>                                | Open a new PRACTICE debug window (command <b>WinResist.WinExt.PLIST</b> ).   |
| <b>Macros</b>                              | Display the PRACTICE stack (command <b>PMACRO.list</b> ).  |
| <b>Edit</b>                                | Open PRACTICE editor PEDIT to edit the PRACTICE script (command <b>WinResist.WinExt.PEDIT</b> ).   |
| <b>Breakpoints</b>                         | Open a <b>PBREAK.List</b> window to display all PRACTICE breakpoints.  |




PRACTICE breakpoint  
in this line

PRACTICE debug-pull-down

| <b>PRACTICE debug-pull-down</b> |   |
|---------------------------------|---|
| <b>Goto Till</b>                | Run PRACTICE script until the selected line (command <b>CONTINUE</b> <line_number>).                    |
| <b>Set PC Here</b>              | Set PRACTICE PC to the selected line.   |
| <b>Breakpoint ...</b>           | Open <b>PBREAK.Set</b> dialog to configure PRACTICE breakpoint.   |
| <b>Toggle breakpoint</b>        | Toggle PRACTICE breakpoint.   |
| <b>Disable breakpoint</b>       | Disable PRACTICE breakpoint (command <b>PBREAK.DISABLE</b> ).   |
| <b>Edit Here</b>                | Open PRACTICE editor to edit the PRACTICE script. The cursor is automatically set to the selected line. |


PRACTICE breakpoints can be set:

- to a specific line of a specified script
- to a specific line in any script (\*)



## Display the PRACTICE Stack

Display the PRACTICE-Stack



The PRACTICE stack displays the program nesting and the PRACTICE macros.

## Run Through Program and Generate a Test Report

---

**Task of part 1 of the script:** Start the program execution and wait until the program execution is stopped at the entry of the function main.

```
// Script run_through_code.cmm

// Part 1

// Prepare debugging
DO "target_setup.cmm"

Go main
WAIT !STATE.RUN() 2.s

IF STATE.RUN()
(
    Break
)
...
```

The script consists of:

### TRACE32 commands

|                     |  |
|---------------------|--|
| <b>Go</b> <address> | Start the program execution.<br>Program execution should stop when <address> is reached. |
| <b>Break</b>        | Stop the program execution.  |

## PRACTICE commands

**DO** <filename>

Call PRACTICE script <filename>

**WAIT** <condition> <time\_period>

Wait until <condition> becomes true or <time\_period> expired.

**IF** <condition>

(

<if\_block>

)

**ELSE**

(

<else\_block>

)

Execute <if\_block> when <condition> is true.

Execute <else\_block> when <condition> is false.

PRACTICE is whitespace sensitive. There must be at least one space after a PRACTICE command word.

<block> has to be set in round brackets. PRACTICE requires that round brackets are typed in a separate line.

## TRACE32 function

**STATE.RUN()**

Returns TRUE when the program execution is running.

Returns FALSE when the program execution is stopped.

**Task of part 2 of the script:** Check if program execution stopped at entry to function main and generate a test report.

```
// Part 2
...
IF Register(PC)==ADDRESS.OFFSET(main)
(
    APPEND "test_protocol.txt" FORMAT.STRING("System booted successfully",70.,' ')\
    FORMAT.UnixTime("c",DATE.UnixTime(),DATE.utcOffset())
)
ELSE
(
    APPEND "test_protocol.txt"  FORMAT.STRING("Bootng failed",70.,' ')\
    FORMAT.UnixTime("c",DATE.UnixTime(),DATE.utcOffset())
    ENDDO
)
```

The backslash \ in conjunction with at least one space serve as a line continuation character.


### TRACE32 function

// Returns the contents of the specified core register as a hex. number.  
**Register**(<register\_name>)

### Addresses in TRACE32

Why is the following function needed?


ADDRESS.OFFSET(main)






```
PRINT Register(PC)
```

```
// print the content of the program  
// counter as a hex. number
```



```
PRINT main
```


```
// print the address of main
```



**main** is an address and addresses in TRACE32 PowerView consist of:

- An **access class** (here P:) which consists of one or more letters/numbers followed by a colon (:)
- A hex. number (here 0x4000105C) that determines the actual address

```
PRINT ADDRESS.OFFSET(main)
```



**ADDRESS.OFFSET**(<symbol>)

Returns the hex. number part of an address.

```
// Part 2
...
IF Register(PC)==ADDRESS.OFFSET(main)
(
  APPEND "test_protocol.txt" FORMAT.STRING("System booted successfully",70.,' ')\
  FORMAT.UnixTime("c",DATE.UnixTime(),DATE.utcOffset())
)
ELSE
(
  APPEND "test_protocol.txt"  FORMAT.STRING("Bootng failed",70.,' ')\
  FORMAT.UnixTime("c",DATE.UnixTime(),DATE.utcOffset())
  ENDDO
)
```

## PRACTICE commands

**APPEND** *<filename>* {*<data>*}

Append data to content of file *<filename>*.

**ENDDO**

A script ends with its last command or with the ENDDO command.

## TRACE32 functions

// Formats *<string>* to the specified *<width>*. If the length of *<string>* is shorter the *<width>*,  
// the *<filling\_character>* is appended.

**FORMAT.STRING**(*<string>*,*<width>*,*<filling\_character>*)

//Returns the time in UNIX format and that is seconds passed since Jan 1st 1970.

**DATE.UnixTime**()

// Returns offset of the local time to UTC in seconds.

**DATE.utcOffset**()

// Format Unix time according to ISO 8601

**FORMAT.UnixTime**("c",DATE.UnixTime(),DATE.utcOffset())


```
// Prepare debugging
DO "target_setup.cmm"

Go main

WAIT !STATE.RUN() 2.s

IF STATE.RUN()
(
    Break.direct
)
IF Register(PC)==ADDRESS.OFFSET(main)
(
    APPEND "test_protocol.txt" FORMAT.STRING("System booted successfully",70.,' ')\
    FORMAT.UnixTime("c",DATE.UnixTime(),DATE.utcOffset())
)
ELSE
(
    APPEND "test_protocol.txt"  FORMAT.STRING("Booting failed",70.,' ')\
    FORMAT.UnixTime("c",DATE.UnixTime(),DATE.utcOffset())
    ENDDO
)
...
```

Results for example in:




# Check Contents of Addresses

**Task of the script:** After an appropriate program address e.g. main is reached, you can check if certain memory addresses are initialized with their correct value.

```
// Script check_memory_locations.cmm
...
IF Data.Long(D:0x40004058) != 0x0
(
  APPEND "test_protocol.txt" \
  FORMAT.STRING("Initialization of address D:0x40004058 failed", 70., ' ') \
  FORMAT.UnixTime("c", DATE.UnixTime(), DATE.utcOffset())
)

IF Data.Long(ANC:0xC3FDC0C4) != 0x0
(
  APPEND "test_protocol.txt" \
  FORMAT.STRING("Initialization of Global Status Register failed", 70., ' ') \
  FORMAT.UnixTime("c", DATE.UnixTime(), DATE.utcOffset())
)
...
```

Results for example in



## TRACE32 function

**Data.Long(<address>)** Returns the contents of the specified address as an 32-bit hex. value.

<address> requires an **access class**.

```
Data.Long(D:0x40004058)           // D: indicates the generic access
                                   // class Data

Data.LONG(ANC:0xC3FDC0C4)          // ANC: indicates
                                   // physical address (A)
                                   // No Cache (NC)
```


# Check Contents of Address Range

**Task of the script:** After an appropriate program address e.g. main is reached, you can check if a certain memory range is initialized with their correct values. An easy way to provide the correct values is a binary file.

```
// Script check_memory_range.cmm
...
Data.LOAD.Binary "range_correct" 0x40004000 /DIFF
IF FOUND()
(
  PRIVATE &s
  APPEND "test_protocol.txt" \
  FORMAT.STRING("Initialization of 0x40004000--0x4000401F failed ",70.,' ')\
  FORMAT.UnixTime("c",DATE.UnixTime(),DATE.utcOffset())

  &s=TRACK.ADDRESS()
  APPEND "test_protocol.txt" \
  FORMAT.STRING("First difference found at: &s",70.,' ')
)
...
```

Results for example in:



## TRACE32 command

**Data.LOAD.Binary** <filename> <address> /DIFF

Compare memory content at <address> with contents of <filename> and provide the result by the following TRACE32 functions:

**FOUND()**

**TRACK.ADDRESS()**

## TRACE32 functions

**FOUND()**

Returns TRUE if a difference was found.

**TRACK.ADDRESS()**

Returns the address of the first difference.

**PRIVATE** {<macro>}

Creates a private PRACTICE macro.

PRACTICE macros start with **&** to make them different from variables from the program under debug.

Private PRACTICE macros are only visible inside the declaring block and are erased when the block ends.


In this example the declaring block are the instructions set in round brackets.

To inspect all differences in detail the following script can be helpful.

```
// Script check_memory_range_details.cmm

Data.LOAD.Binary "range_correct" VM:0x40004000
Data.Dump VM:0x40004000++0x1f /SpotLight
SCREEN.display
Data.COPY 0x40004000++0x1f VM:0x40004000
```

Results for example in:



## TRACE32 commands

**Data.LOAD.Binary** <filename> VM:<address>

Load the contents of <filename> to <address> in the **TRACE32 virtual memory**. The TRACE32 Virtual Memory is memory on the host computer which can be displayed and modified with the same commands as real target memory.

**Data.dump** VM:<address\_range> /SpotLight

Display the contents of the TRACE32 Virtual Memory for the specified <address\_range>.

The option **SpotLight** advises TRACE32 to mark changed memory locations if the window is displayed in TRACE32 PowerView.

**Data.COPY** <address\_range> VM:<address>

Copy the content of <address\_range> to the TRACE32 Virtual Memory.

## PRACTICE commands

If PRACTICE scripts are executed, the screen is only updated after a PRINT command.

**SCREEN.display**

Advise TRACE32 to update the screen now.


# Check the Contents of Variables

**Task of the script:** After an appropriate program address e.g. main is reached, you can check if certain variables are initialized with their correct value.

```
// Script check_var.cmm
...
Var.IF stra2[1][0].pastruct5[0]!=0.
(
  APPEND "test_protocol.txt"\
  FORMAT.STRING("Initialization of stra2[1][0].pastruct5[0] failed",70.,' ')\
  FORMAT.UnixTime("c",DATE.UnixTime(),DATE.utcOffset())

  APPEND "test_protocol.txt" "stra2[1][0].pastruct5[0]= "\
  %Decimal Var.Value(stra2[1][0].pastruct5[0])
)
...
```

My result for example in:



## PRACTICE command

```
Var.IF <hll_condition>
(
  <block>
)
```

Execute <block> when the condition written in the programming language used is true (C, C++, ...)

## PRACTICE function

```
Var.VALUE(<hll_expression>)
```

Returns the contents of the variable/variable component specified by <hll\_expression> as a hex. number.



# Record Formatted Variables

**Task of script:** Write the content of various variables to a file. Use the same formatting as **Var.View** command.


```
// Script record_var.cmm
...
PRinTer.FileType ASCIIIE
PRinTer.OPEN "test_var.lst"

WinPos ,,,,0
WinPrint.Var.View %String cstr1

WinPos ,,,,0
WinPrint.Var.View %Open str2

WinPos ,,,,0
WinPrint.Var.View vfloat

PRinTer.CLOSE
...
```



## TRACE32 commands


|                                  |   |
|----------------------------------|---|
| <b>PRinTer.FileType</b> <format> | Specify output <format> for output file.  |
| <b>PRinTer.Open</b> <filename>   | Open file <filename> for outputs.   |
| <b>PRinTer.CLOSE</b>             | Close open output file.   |
| <b>WinPrint.</b> <command>       | Redirect the <command> output to the specified file.  |
| <b>WinPos</b> ,,,,0              | By the default the TRACE32 <command> and its output is redirected to the specified file.<br>With this special WINPOS command only the <command> output is redirected to the specified file. |

**Task of the script:** Write the contents of the variable `vbfield` to a file whenever the program execution stops at the specified breakpoint. Use CSV as output format.

```
// Script test_var_vbfield.cmm
...
Break.RESet
Var.Break.Set vbfield /Write

REPEAT 10.
(
    Go
    WAIT !STATE.RUN() 2.s
    IF STATE.RUN()
    (
        Break
        ENDDO
    )
    Var.EXPORT "vbfield_export.csv" vbfield /Append
)
...
```

Results for example in:



## TRACE32 commands

```
// Use expression of your programming language (C, C++, ...) to specify write breakpoint
Var.Break.Set <hdl_expression> /Write

// Since the number of write breakpoints is limited, it is recommended to reset the current breakpoint
// settings
Break.RESet

// Append content of variables as CSV (Comma Separated Values) to file <filename>
Var.EXPORT <filename> [{%<format>}] {<variable>} /Append
```

## PRACTICE command

```
RePeaT <count>           Repeat <block> <count>-times.
(
  <block>
)
```

**Task of the script:** Test functions with specified parameters and generate a test protocol.


```
// Script test_function.cmm
...
PRIVATE &result

&result=FORMAT.DECIMAL(8.,Var.VALUE(func5(2,6,9)))
APPEND "test_protocol.txt" \
FORMAT.STRING("func5(2,6,9) tested with result &result",70.,' ') \
FORMAT.UnixTime("c",DATE.UnixTime(),DATE.utcOffset())

&result=FORMAT.DECIMAL(8.,Var.VALUE(func5(14,87,93)))
APPEND "test_protocol.txt" \
FORMAT.STRING("func5(14,87,93) tested with result &result",70.,' ') \
FORMAT.UnixTime("c",DATE.UnixTime(),DATE.utcOffset())

&result=FORMAT.DECIMAL(8.,Var.VALUE(func5(44,44,44)))
APPEND "test_protocol.txt" \
FORMAT.STRING("func5(44,44,44) tested with result &result",70.,' ') \
FORMAT.UnixTime("c",DATE.UnixTime(),DATE.utcOffset())
...
```

Results for example in:



## PRACTICE function

**FORMAT.DECIMAL**(*<width>*,*<value>*)

Formats a numeric expression to a decimal number and generates an output string with a fixed length of *<width>* with leading spaces. Numeric expressions which need more characters than *<width>* for their loss-free representation aren't cut.

**Var.VALUE**(*<function\_call>*)

Returns the return value of called function as hex. number.

# Test Function with Parameter File

**Task of script:** Test functions, but provide function name, parameters and expected result by a parameter file. Generate a test protocol.

```
// Script test_func_param.cmm

LOCAL &testfunc &correct_result

OPEN #1 "func_test.txt" /READ

WHILE TRUE()
(
  READ #1 &testfunc &correct_result
  IF "&testfunc"!=" "
  (
    GOSUB perform_test
  )
  ELSE
  (
    CLOSE #1
    ENDDO
  )
)
ENDDO

perform_test:
(
  IF Var.VALUE(&testfunc)==&correct_result
  (
    APPEND "test_protocol.txt"\
    FORMAT.STRING("&testfunc=&correct_result",50.,' ')\
    FORMAT.UnixTime("c",DATE.UnixTime(),DATE.utcOffset())
  )
  ELSE
  (
    PRIVATE &result
    &result=CONVERT.HEXTOINT(Var.VALUE(&testfunc))
    APPEND "test_protocol.txt"\
    FORMAT.STRING("&testfunc failed with &result (&correct_result)",50.,' ')\
    FORMAT.UnixTime("c",DATE.UnixTime(),DATE.utcOffset())
  )
  RETURN
)
```

Example for a parameter file.

```
func5(22,12,17) 56.
func5(14,87,93) 8105.
func5(44,44,44) 1980.
```

Results for example in:

## PRACTICE command

### **GOSUB** <label>

Call a subroutine. The start of the subroutine is identified by <label>.

Labels must start in the first column of a line and end with a colon. No preceding white space allowed.

Subroutines are usually located after the ENDDO statement.

### **RETURN**

Return from subroutine.

### **LOCAL** <macro>

Creates a local PRACTICE macro.

Local PRACTICE macros are visible inside the declaring block, in all called scripts and in all called subroutines.

They are erased when the declaring block ends. The declaring block is here the script itself.

### **WHILE** <condition>

Execute <block> as long as <condition> is true.

```
(  
  <block>  
)
```

### **OPEN** #<buffer\_number> <filename> /Read

Open file <filename> for reading. The file is referenced by its #<buffer\_number> by the following commands.

### **READ** #<buffer\_number> {<macro>}

Read next line from file referenced by #<buffer\_number> into PRACTICE macros.

Space serves as parameter separators.

### **CLOSE** #<buffer\_number>

Close file referenced by #<buffer\_number>.

## TRACE32 function

### **CONVert.HEXTOINT**(<number>)

Convert <number> to a decimal number.

## Program Elements

---

### Commands

---

All commands of the TRACE32 development tools

Commands for program flow control and conditional commands


I/O commands

# Functions


Functions are used to get information about the state of the target system or the state of the development tool.

|  |  |
|--|--|
| <b>Register(&lt;name&gt;)</b>          | Get the content of the specified CPU register.   |
| <b>V.VALUE(&lt;hll expression&gt;)</b> | Get the contents of a hll expression.  |
| <b>STATE.RUN()</b>                     | Returns true if program is running on the target, returns false if program execution is stopped. |
| <b>OS.PWD()</b>                        | Returns the name of the current working directory as a string.                                   |
| <b>CONV.CHAR(&lt;value&gt;)</b>        | Converts an integer value to an ASCII character.   |

A list of all available functions can be found in the on-line help:



Select functions only





**CPU()**

Returns the name of the selected processor as a string

```
PRINT CPU()

IF (CPU()=="TC1796")
    &int_flsh_size=0x00200000
ELSE IF (CPU()=="TC1766")
    &int_flsh_size=0x00178000
...
```

**SYSTEM.UP()**

Returns TRUE if the communication between the debugger and the CPU is active

**STATE.RUN()**

Returns TRUE if the CPU is executing the application program

```
IF !SYSTEM.UP()
    SYStem.Up

IF STATE.RUN()
    BREAK
```

**VERSION.BUILD()**

Returns the build number

```
IF (VERSION.BUILD())<1146.)
    PRINT %ERROR "The version of TRACE32 too old!"
```

**Data.Byte (<address>)**

Returns byte at &lt;address&gt;

**Data.Word (<address>)**

Returns word at &lt;address&gt;

**Data.Long (<address>)**

Returns long at &lt;address&gt;

```
PRINT Data.Long(D:0xd00000008)
```

```
; script function_string.cmm

DIALOG.FILE *.sre
ENTRY &filename
&filename=STRING.UPR("&filename")

PRINT "&filename"

ENDDO
```

## Comments

---

Comments start with “/” or “;”

The “;” has a special meaning with the **Var** command group.

```
Var.View flags[3];ast.count;i
```

Since the command **SETUP.Var %SPace.ON** allows spaces in variable expressions, “;” operates as a separator of variable expressions.

## Labels

---

Labels have to be entered at the first column and they end with “:”



It is recommended to avoid unnecessary blanks. Unnecessary blanks can lead to a misinterpretation of PRACTICE commands.

```
&i = 7.
```

```
; unnecessary blanks can lead to  
; misinterpretations
```

```
&i=7.
```

Start a PRACTICE Program in another PRACTICE Program

A PRACTICE program can call another PRACTICE program. This allows you to structure your PRACTICE programs.

**DO** <filename> [<parlist>]                      Start a PRACTICE program in another PRACTICE program

```

; PRACTICE file modular.cmm

AREA.Create IO-AREA                      ; Create and open IO window
AREA.Select IO-AREA
WinPOS , , , , , IO1
AREA.view IO-AREA


DO iotext                                ; Start PRACTICE script iotext

AREA.RESet                               ; Close IO window
WinCLEAR IO1

ENDDO
```

The following abbreviations support the writing of movable PRACTICE scripts:

| Windows | function   |
|---------|--|
| .\      | current directory                                      |
| ..\     | parent directory                                       |
| ~\      | home directory of user (from \$HOME)                   |
| ~~\     | system directory of TRACE32                            |
| ~~~\    | temporary directory for TRACE32                        |
| ~~~~\   | directory where the current PRACTICE script is located |



PRACTICE manages the PRACTICE program call hierarchy on a PRACTICE stack. Local and global PRACTICE variables are also maintained on this stack.



Each PRACTICE program should end with an **ENDDO** instruction.

**ENDDO** [*<param>*]

Return from PRACTICE program

With the return from the PRACTICE program a result parameter can be passed.

```
; script test_status
```

```
ENDDO (0==0)
```

```
; return TRUE as result
```

```
; ENDDO (1==0)
```

```
; return FALSE as result
```

```
; script enddo_param
```

```
DO setup_CPU
```

```
DO test_status
```

```
ENTRY &result
```

```
; Read result
```

```
IF &result
```

```
; Evaluate result
```

```
    DIALOG.OK "Test passed"
```

```
ELSE
```

```
(
```

```
    DIALOG.OK "Test failed"
```

```
    ENDDO
```

```
)
```

```
ENDDO
```

|  |   |
|--|---|
| <b>GOSUB</b> <i>&lt;label&gt;</i> [ <i>&lt;parlist&gt;</i> ] | Call a PRACTICE subroutine                        |
| <b>RETURN</b> [ <i>&lt;parlist&gt;</i> ]                     | Return from a PRACTICE subroutine                 |
| <b>GOTO</b> <i>&lt;label&gt;</i>                             | Branch within the same PRACTICE program           |
| <b>JUMPTO</b> <i>&lt;label&gt;</i>                           | Branch to a label within another PRACTICE program |

### ENTRY <parlist>

The ENTRY command can be used to

- Pass parameters to a PRACTICE program or to a subroutine
- To return a value from a subroutine

#### Example 1: Pass parameters to a PRACTICE program

```
; Practice program patch.cmm
; DO patch.cmm 0x1000++0xff 0x01 0x0a

ENTRY &address_range &data_old &data_new

IF "&address_range"==" "
(
    PRINT "Address range parameter is missing"
    ENDDO
)

IF "&data_old"==" "
(
    PRINT "Old data parameter is missing"
    ENDDO
)

IF "&data_new"==" "
(
    PRINT "New data parameter is missing"
    ENDDO
)

Data.Find &address_range &data_old

IF FOUND()
(
    Data.Set TRACK.ADDRESS() &data_new
    Data.Print TRACK.ADDRESS()
    DIALOG.OK "Patch done"
)
ELSE
    DIALOG.OK "Patch failed"

ENDDO
```

**Example 2:** Pass parameters to a subroutine and get back the return value.

Pass parameter to subroutine

```
; Practice-program param.cmm
```

```
GOSUB test sieve++0x20 0x01
```

```
ENTRY &found
```

```
PRINT "Data found at address " &found
```

```
ENDDO
```

```
test:
```

```
ENTRY &address_range &data
```

```
Data.Find &address_range &data
```

```
&result=ADDRESS.OFFSET(TRACK.ADDRESS())
```

```
RETURN &result
```

Get back the return value



# Conditional Program Execution

---

For detailed information on logical operations refer to [chapter “Operators”](#) in “[IDE User’s Guide](#)” ([ide\\_user.pdf](#)).

## IF/ELSE

---

```
IF <condition>  
    <block>  
[ELSE  
    <block>]
```

### Example:

```
; PRACTICE program patch.cmm  
  
ENTRY &address_range &data_old &data_new  
  
Data.Find &address_range &data_old  
  
IF FOUND()  
(  
    Data.Set TRACK.ADDRESS() &data_new  
    Data.Print TRACK.ADDRESS()  
    DIALOG.OK "Patch done"  
)  
ELSE  
    DIALOG.OK "Patch failed"  
ENDDO
```

Block structure is similar to C.

“(“ and “)” have to be in a separate line.

**IF must** be followed by space.

```
Var.IF <hll-condition>  
  <block>  
[ELSE  
  <block>]
```

**Example:**

```
Var.IF (flags[0]==flags[5])  
  PRINT "Values are equal"  
ELSE  
  PRINT "Values are not equal"  
ENDDO
```

```
WHILE [<condition>]  
  <block>  
Var.WHILE [<hll-condition>]  
  <block>  
RePeaT [<count>]  
  <block>  
RePeaT  
  <block>  
[WHILE [<condition>]]
```

Macros are the variables of the script language PRACTICE. They work as placeholders for a sequence of characters.

Macro names in PRACTICE always start with an **&** sign, followed by a sequence of letters (a-z, A-Z) and numbers (0-9); you can also use the **\_** symbol in macro names. It is recommended not to use macro names which start with a number after the initial **&** sign.

Macro names are case sensitive, so **&a** is different from **&A**.

## Create a Macro

---

Empty PRACTICE macros **can** be created along with their scope by using one of the following PRACTICE commands:

### **PRIVATE** {<macro>}

Create PRIVATE macros.

PRIVATE macros are visible in the script, subroutine or block in which they are created. And they are removed when the script, subroutine or block ends.

They are visible in nested blocks, but not in called subroutines and called scripts.

### **LOCAL** {<macro>}

Create LOCAL macros.

LOCAL macros are visible in the script, subroutine or block in which they are created. And they are removed when the script, subroutine or block ends.

They are visible in nested blocks, in called subroutines and called scripts.

### **GLOBAL** {<macro>}

Create GLOBAL macros.

GLOBAL macros are visible to all scripts, subroutines and blocks until they are explicitly removed by the command **PMACRO.RESet**. They can only be removed if no script is running.

```
// Script test_func_param.cmm

//LOCAL &testfunc &correct_result

OPEN #1 "func_test.txt" /READ

WHILE TRUE()
(
  READ #1 &testfunc &correct_result
  IF "&testfunc"!=" "
  (
    GOSUB perform_test
  )
  ELSE
  (
    CLOSE #1
    ENDDO
  )
)
ENDDO

perform_test:
(
  IF Var.VALUE(&testfunc)==&correct_result
  (
    APPEND "test_protocol.txt"\
    FORMAT.STRING("&testfunc=&correct_result",50.,' ')\
    FORMAT.UnixTime("c",DATE.UnixTime(),DATE.utcOffSet())
  )
  ELSE
  (
    PRIVATE &result
    &result=CONVERT.HEXTOINT(Var.VALUE(&testfunc))
    APPEND "test_protocol.txt"\
    FORMAT.STRING("&testfunc failed with &result (&correct_result)",50.,' ')\
    FORMAT.UnixTime("c",DATE.UnixTime(),DATE.utcOffSet())
  )
  RETURN
)
```

The PRACTICE interpreter will create a new LOCAL macro when an assignment is done, but it cannot find a macro of that name in the current scope.

The command **PMACRO.EXPLICIT** advises the PRACTICE interpreter to generate an error if a PRACTICE macro is used in an assignment, but was not created in advance. It also advises the PRACTICE interpreter to generate an error if the same macro was intentionally created a second time within its scope.

## Assign Content to a Macro

```
// Script

PRIVATE &my_string &my_range &my_var
PRIVATE &my_number &my_float &my_var_value &my_expression
PRIVATE &my_boolean &my_boolean_result


// Assign a string
&my_string="Hello World"
&my_range="0x40004000++0xffff"
&my_var=Var.STRING(cstr1)

// Assign a numbers
&my_number=100.
&my_float=5.667e13
&my_var_value=Var.VALUE(ast.count)
&my_expression=&my_number*&my_var_value

// Assign a boolean
&my_boolean=TRUE()
&my_boolean_result=STATE.RUN()


ENDDO
```

After a content is assigned to a macro, tooltips allows to inspect its current content.



The PRACTICE stack provide an overview for all macros.

The keyword PRIVATE is used to identify PRIVATE macros.



**Var.STRING(<hll\_expression>)**

Returns a zero-terminated string, if <hll\_expression> is a pointer to character or an array of characters. Returns a string that represents the variable contents otherwise.

To better understand the usage of macros, it is the best to look at the way the PRACTICE interpreter executes a script line.

The PRACTICE interpreter executes a script line by line. Each line is (conceptually) processed in three steps:

1. All macros are replaced by their corresponding character sequence.
2. All expressions are evaluated.
3. The resulting command is executed.

For the following examples the command **PRINT** is used.

**PRINT** {<data>}


PRINT specified <data> to TRACE32 Message Line and to TRACE32 Message AREA.

### Example 1

```
PRIVATE &a  
  
&a="5+4 "  
PRINT &a
```

We look at the **PRINT &a** line. To execute this line, the interpreter will first:

1. Replace all macros with their corresponding character sequences. So:  
**PRINT &a -> PRINT 5+4**
2. Evaluate all expressions  
**PRINT 5+4 -> PRINT 9**
3. Execute the resulting command. So it will execute **PRINT 9**.




## Example 2

```
PRIVATE &a

&a="Hello World"
PRINT &a
```

This example will generate an error.




Let's look at the three steps the interpreter will take to execute the `PRINT &a` command:

1. Replace all macros with their corresponding character sequences  
**`PRINT &a` -> `PRINT Hello World`**
2. Evaluate expressions  
**`PRINT Hello World` -> **error****
3. Execute command, which will not happen because of the error in the second step.

The second step fails because in PRACTICE a single word like **Hello** (which is not enclosed in double quotes) refers to a debug symbol, loaded for example from an ELF file.

When the PRACTICE interpreter encounters such a debug symbol, the expression evaluation will try to replace the debug symbol by the address to which the symbol refers. If there is no debug symbol called **Hello** (which is likely), the PRACTICE interpreter will output the error message **symbol not found**.

If by pure accident there are debug symbols called **Hello** and **World** the addresses of these symbols will be printed.




This example demonstrates how the pure macro replacement step will basically always work, since you always can replace a macro by its character sequence; but the result might not make sense.



Macros are replaced by their character sequence. If you want to explicitly use this character sequence as a string, then you should enclose the macro in double quotes, for example:

```
PRIVATE &a  
  
&a="Hello World"  
PRINT "&a"
```



To understand what happens it is again best to look at the three steps which are taken to execute the **PRINT "&a"** command.

1. Replace the macro by its character sequences  
**PRINT "&a" -> PRINT "Hello World"**
2. Evaluate expressions.  
Nothing to do for this example.
3. Execute command.

## String composing example:

```
// Script string_example.cmm

PRIVATE &drive &architecture &demo_directory
&drive="C:"
&arch="MPC"

// PRINT command
PRINT "Directory " "&drive" "\T32_" "&architecture" "\demo"
PRINT "Directory "+"&drive"+"&T32_"+"&architecture"+"&demo"
PRINT "Directory &(drive)\T32_&(architecture)\demo"

// Macro assignment
&demo_directory="&drive"+"&T32_"+"&architecture"+"&demo"
DIR &demo_directory

&demo_directory="&(drive)\T32_&(architecture)\demo"
DIR &demo_directory

// Command parameter
DIR "&(drive)\T32_&(architecture)\demo"
```

**DIR** <directory>

Display a list of files and folders for the specified directory.

```
// Script numbers.cmm

PRIVATE &my_hex &my_dec &my_bin
PRIVATE &my_stringlength &my_sizeof
PRIVATE &add1 &add2
PRIVATE &convert1 &convert2

// Hex, decimal, binary by TRACE32 syntax
&my_hex=0x7
&my_dec=22.
&my_bin=0y1110

// Hex, decimal, binary as expression result
&add1=&my_bin+&my_hex
&add2=&my_hex+&my_dec


// Hex, decimal, binary as return value
&my_stringlength=STRing.LENgtH("0123456789012345")
&my_sizeof=sYmbol.SIZEOF(sieve)

// Hex, decimal, binary by CONVERT function

&convert1=CONVERT.HEXTOINT(&my_bin)
&convert2=CONVERT.HEXTOINT(&my_hex)

...
```

The PRACTICE stack shows the macro values and their radix.



**STRing.LENgtH(<string>)**

Returns the length of the <string> as a decimal number.

**sYmbol.SIZEOF(<symbol>)**

Returns the size occupied by the specified debug <symbol> (e.g. function, variable, module) as a hex. number.

But if you use a PRACTICE output command, the radix information is removed.


```
// Script append_example.cmm
&PRIVATE &target_id
&target_id="D:0x40004000"

DEL "my_append.txt"

APPEND "my_append.txt" "Software tested with:"
APPEND "my_append.txt" " "
APPEND "my_append.txt" "Debug Cable: " CABLE.SERIAL()
APPEND "my_append.txt" "TRACE32 Version: " VERSION.BUILD()
APPEND "my_append.txt" "Target Version: " Data.Long(&target_id)

// ...
```

Results for example in:



## TRACE32 command

|                       |                                     |
|-----------------------|-------------------------------------|
| <b>DEL</b> <filename> | Delete file specified by <filename> |
|-----------------------|-------------------------------------|

## TRACE32 functions

|                        |   |
|------------------------|---|
| <b>CABLE.SERIAL()</b>  | Returns the first serial number of the plugged debug cable.   |
| <b>VERSION.BUILD()</b> | Returns build number of TRACE32 software as a decimal number. |


Since it might be confusing for the reader not to know if a number is decimal or hex, you can proceed as follows:

```
// Script append_example_format.cmn

&PRIVATE &target_id
&target_id="D:0x40004000"
DEL "my_append.txt"

APPEND "my_append.txt" "Software tested with:"
APPEND "my_append.txt" " "
APPEND "my_append.txt" "Debug Cable:  " FORMAT.STRING(CABLE.SERIAL(),15.,' ')
APPEND "my_append.txt" "TRACE32 Version:  " FORMAT.DECIMAL(8.,VERSION.BUILD())+"."
APPEND "my_append.txt" "Target Version: 0x"+FORMAT.HEX(8.,Data.Long(&target_id))
```

May result for example in:



## TRACE32 function

**FORMAT.HEX(<width>,<number>)**

Formats a numeric expression to a hexadecimal number and generates an output string with a fixed length of <width> with leading zeros.

PRACTICE macros are not available in the command line. They are only available when running a script. But you can proceed as follow to test a macro assignment:



For all complex data structures TRACE32-internal variables can be used. The following two commands can be used to declare a TRACE32-internal variable.

**Var.NEWGLOBAL** <type> \<name> Create a global TRACE32-internal variable

**Var.NEWLOCAL** <type> \<name> Create a local TRACE32-internal variable

TRACE32-internal variables require that a program is loaded via the **Data.LOAD** command. All data types provided by this program can then be used (**sSymbol.List.Type**).

- TRACE32-internal variables have the same scope as PRACTICE macros (e.g. they are on the PRACTICE stack).
- TRACE32-internal variables are displayed and modified via the **Var** command group.

```
; script newlocal.cmm

LOCAL &my_symbol
ENTRY &my_symbol

Var.NEWLOCAL char[5][40] \typeresult

Var.Assign \typeresult[0]="Symbol does not exist"
Var.Assign \typeresult[1]="Symbol is label"
Var.Assign \typeresult[2]="Symbol is function"
Var.Assign \typeresult[3]="Symbol is variable"
Var.Assign \typeresult[4]="Undefined"

&n=sSymbol.TYPE(&my_symbol)
Var.PRINT %String \typeresult[&n]
ENDDO
```

**Var.Assign** %<format> <variable> Modify variable, no log is generated in the message line and the **AREA** window.

**sSymbol.TYPE**(<symbol>) Returns the basic type of the symbol as a numerical value.

0 = symbol does not exist  
1 = plain label without type information  
2 = HLL function  
3 = HLL variable  
other values may be defined in the future

## Output Command

---


**PRINT** *<format> <parlist>*

**DIALOG.OK** *<message>*

```
PRINT "FLASH programmed successfully"
```

```
PRINT %ERROR "FLASH programming failed"
```

```
DIALOG.OK "Patch done"
```






|                               |                              |
|-------------------------------|------------------------------|
| <b>ENTER</b> <parlist>        | Window based input           |
| <b>INKEY</b> [<par>]          | Input command (character)    |
| <b>DIALOG.YESNO</b> <message> | Create a standard dialog     |
| <b>DIALOG.File</b> <message>  | Read a filename via a dialog |

```
INKEY                                ; Wait for any key

INKEY &key                             ; Wait for any key, key
                                       ; code is entered to &key
```

```
; PRACTICE program dialog.cmm
DIALOG.YESNO "Continue with hardware test?"
```



```
ENTRY &result

IF &result
(
    PRINT "Test started"
    DO test2
)
ELSE
    PRINT "Test aborted"
ENDDO
```

```
DIALOG.File *sre
ENTRY &filename

Data.LOAD.S3record &filename

ENDDO
```

An I/O window is needed for PRACTICE inputs and outputs. To realize this an AREA window is used.

### Open and assign an AREA window

1. Create an AREA-Window

```
AREA.Create [<area> ]
```

2. Select an AREA window for PRACTICE I/O.

```
AREA.Select [<area>]
```

3. Select the screen position of the AREA window. This command is used here, because it allows you to assign a name to an AREA window. This is useful, if you want to delete this window after the I/O procedure.

```
WinPOS [<pos>] [<size>] [<scale>] [<name>] [<state>] [<header>]
```

4. Display AREA-Window

```
AREA.view [<area>]
```

### Remove AREA Window

1. Resets the AREA window settings to the default settings: the message area (AREA A000) is used for error and system messages. No other AREA window is active.

```
AREA.RESet
```

2. Delete a specific windows.

```
WinCLEAR [<pagename>|<windowname>|TOP]
```

```
; PRACTICE file iowindow.cmm

AREA.Create IO-AREA
AREA.Select IO-AREA
WinPOS , , , , , IO1
AREA.view IO-AREA

PRINT "Please enter the address"
PRINT "Address="
ENTER &a
PRINT " "
PRINT "Entered address=" &a

WAIT 2.s

AREA.RESet
WinCLEAR IO1

ENDDO
```


It is also possible to use the DIALOG programming feature to realize an I/O.

```
; PRACTICE file dialog_prog.cmm

DIALOG
(
    HEADER "Address entry"
    POS 0. 0. 25.
    TEXT "Address"
    POS 1. 1. 10.
ADD: DEFEDIT " " " " "
    POS 1. 3. 5.
    DEFBUTTON "OK" "GOTO okclose"
)
STOP

okclose:
    &address=DIALOG.STRING(ADD)
    PRINT "Entered address=" &address

DIALOG.END
ENDDO
```



### STOP

Stop the execution of a PRACTICE program

In the example above the execution of the PRACTICE program is continued with the **GOTO okclose** command, that is assigned to the DEFBUTTON.

```

DIALOG
(
    HEADER "Select TriCore"                ; Dialog header
    POS 0. 0. 25. 1.                      ; Increase dialog width
    TEXT ""                                ; by empty text
    POS 1. 1. 10.
    TC.1796: CHOOSEBOX "TC1796" ""        ; Define 2 choose boxes
    TC.1766: CHOOSEBOX "TC1766" ""        ; They are exclusive
    POS 1. 4. 5.                          ; alternatives
    DEFBUTTON "OK" "continue"
)

DIALOG.SET TC.1796                        ; Define default setting
                                           ; for choosebox


    STOP
    IF DIALOG.BOOLEAN(TC.1796)             ; Evaluate result
        CPU="TC1796"

    IF DIALOG.BOOLEAN(TC.1766)
        CPU="TC1766"

DIALOG.END

PRINT "Selected CPU= " "&CPU"
ENDDO

```



For more information on the **DIALOG** programming commands refer to *Operation System Reference*.

# File Commands

File commands are used to store data into a file or to read test data from a file

- Open file

**OPEN** #<buffer> <filename> / Read | Write | Create

- Close file

**CLOSE** #<buffer>

- Read data from an open file

**READ** #<buffer> [ %LINE ] <parlist>

- Write data to an open file

**WRITE** #<buffer> <parlist>

## Example 1:

```
; PRACTICE file write.cmm

OPEN #1 abc.txt /Create

&i=0

WHILE &i<10.
(
    WRITE #1 "flags[" &i "]=" V.VALUE("flags[&i]")
    &i=&i+1
)

CLOSE #1

ENDDO
```

**Example 2:** Reads data from a file. Single data are separated by blank or carriage return within the file.

```
; PRACTICE file read.cmm

OPEN #2 read.txt /Read

READ #2 &e1 &e2 &e3 &e4

PRINT "&e1" " " "&e2" " " "&e3" " " "&e4"

CLOSE #2

ENDDO
```

**Example 3:** Read data from a file. Read always a complete line.

```
; PRACTICE file read_line.cmm

OPEN #2 read.txt /Read

READ #2 %LINE &e1

PRINT "&e1"

CLOSE #2

ENDDO
```

|                             |   |
|-----------------------------|---|
| <b>ON ERROR</b> <command>   | Perform commands on PRACTICE runtime error                                      |
| <b>ON SYSUP</b> <command>   | Perform commands when the communication between debugger and CPU is established |
| <b>ON POWERUP</b> <command> | Perform commands on target power on   |

```
...
ON ERROR GOTO
    (
        DIALOG.OK "Abortion by error!"
        ENDDO (0!=0)
    )
...
```

```
ON POWERUP GOTO startup

IF !(STATE.POWER())
    STOP

startup:
    SYStem.CPU TC1796
    WAIT 0.5s
    SYStem.UP

ENDDO
```



